

This is a demo for the following equation.

$$(n_1!n_2!\dots)^{1/2} |n_1, n_2, \dots\rangle = (a_1^\dagger)^{n_1} (a_2^\dagger)^{n_2} \dots |0\rangle$$

For the demo, kets are identified as vectors of harmonic oscillator eigenfunctions.

$$|n_1, n_2, \dots\rangle \mapsto \begin{pmatrix} \psi_{1,n_1} \\ \psi_{2,n_2} \\ \vdots \end{pmatrix} \quad |0\rangle \mapsto \begin{pmatrix} \psi_{1,0} \\ \psi_{2,0} \\ \vdots \end{pmatrix}$$

The eigenfunctions are

$$\psi_{k,n} = \left[\frac{1}{\sqrt{2\pi\sigma_k^2}} \exp\left(-\frac{x^2}{2\sigma_k^2}\right) \right]^{1/2} \frac{1}{\sqrt{2^n n!}} H_n\left(\frac{x}{\sqrt{2}\sigma_k}\right)$$

where $k = 1, 2, \dots$ is the wave number, $n = 0, 1, 2, \dots$ is the energy level, H_n is the n th Hermite polynomial, and

$$\sigma_k^2 = \frac{\hbar}{2m\omega_k}$$

$$\omega_k = \sqrt{k/m}$$

The creation operator acting on an eigenfunction $\psi_{k,n}$ is

$$a_k^\dagger \psi_{k,n} = \frac{x}{2\sigma_k} \psi_{k,n} - \sigma_k \frac{d}{dx} \psi_{k,n} = \sqrt{n+1} \psi_{k,n+1}$$

For example

$$a_k^\dagger \psi_{k,0} = \psi_{k,1}$$

The following script computes

$$A = (n_1!n_2!n_3!)^{1/2} \frac{1}{\sqrt{3}} \begin{pmatrix} \psi_{1,n_1} \\ \psi_{2,n_2} \\ \psi_{3,n_3} \end{pmatrix} \quad B = (a_1^\dagger)^{n_1} (a_2^\dagger)^{n_2} (a_3^\dagger)^{n_3} \frac{1}{\sqrt{3}} \begin{pmatrix} \psi_{1,0} \\ \psi_{2,0} \\ \psi_{3,0} \end{pmatrix}$$

and shows that

$$A - B = \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$$

The factor $1/\sqrt{3}$ is a normalization constant so that

$$\frac{1}{n_1!n_2!n_3!} \int_{-\infty}^{+\infty} A^\dagger A dx = 1$$

```

-- www.eigenmath.org/quantum-harmonic-oscillator-2.txt
omega = sqrt(k/m)
sigma = sqrt(hbar/(2*m*omega))
W = 1/sqrt(2*pi*sigma^2) * exp(-x^2/(2*sigma^2))
psi(k,n) = sqrt(W) * 1/sqrt(2^n*n!) * hermite(x/(sqrt(2)*sigma),n)
-- define creation operator
adag(k,psi) = x / (2*sigma) * psi - sigma * d(psi,x)
-- define demo energy levels
n1 = 1
n2 = 2
n3 = 3
-- compute A
A = sqrt(n1!*n2!*n3!) * (psi(1,n1), psi(2,n2), psi(3,n3)) / sqrt(3)
-- compute B
B = (psi(1,0), psi(2,0), psi(3,0)) / sqrt(3)
for(n,1,n3, B[3]=adag(3,B[3])/sqrt(n), B=sqrt(n)*B)
for(n,1,n2, B[2]=adag(2,B[2])/sqrt(n), B=sqrt(n)*B)
for(n,1,n1, B[1]=adag(1,B[1])/sqrt(n), B=sqrt(n)*B)
-- compare A and B
A - B
-- check normalization
P(f,a,b) = 1/10 * sum(h,10*a,10*b-1,float(eval(f,x,h/10)))
m = 1
hbar = 1
f = dot(conj(A),A) / (n1!*n2!*n3!)
P(f,-10,10)

```